
7. Random walks

[Gould+Tobochnik 7.3, G+T 12]

On part II of these MC courses, “MC simulations in physics”, we deal with actual simulation of

physical processes.

• Let us start with a typical problem in physics, a couple of drunken sailors.

7.1. Introduction

7.1.1. The two drunken sailors

Consider a sailor who gets out of a corner pub completely drunk, in a city where all blocks are in

perfect squares.

• The drunken sailor walks from one corner of a block to the next

• When in a corner, he forgets where he was coming from, and where he is going. He keeps

going, picking the next direction where to go randomly among the four possible ways.

• Then in the following corner he again forgets where he was going, but keeps on walking,

forming a random walking pattern.

Consider another drunken sailor, who has drunk even more, and fallen asleep in the middle of a

huge square.

• He also starts walking, but after a few steps he stops, changes direction randomly.

• Then he takes another few steps, again changes direction, and so on.

We obviously could use Monte Carlo methods to simulate the walk of the two drunken sailors, and

answer questions like how far they on average have come after N steps. But why on earth would

we want to, being physicists?

7.1.2. ...and what this has to do with physics

The answer is that exactly the same kind of random walk processes are surprisingly prevalent in

physics.

• The random walk performed by the sailor walking among the square blocks can e.g.

correspond exactly to the motion of an atom migrating on a (square) lattice in 2D (e.g.

a surface).

• The second case, the movement of the sailor on the square, is migration in a continuum,

and can correspond (in a 3D generalization) e.g. to

– ... atom movement in a liquid

– ... motion of a photon in a star

– ... motion of a neutron in fissile material

Because of such relations to the natural sciences, the basic “drunken sailor” random walk problem

is widely used in many branches of science.

• As we shall soon show, answering e.g. the question of average expected movement

distance for a given number of steps N has a direct correspondence to the definition of

the diffusion constant!

7.2. Simple random walks and diffusion

[G+T 7.3]

7.2.1. One-dimensional walk

Let us first consider the simplest possible case, a random walker in a one-dimensional lattice:

0

Say that a walker begins at x = 0, and that all steps are of equal length l. After each time

interval τ the walker has an equal probability of moving left or right. The direction of each step is

independent of the previous one. Let us denote the displacement at each step by si, for which

si =


+l with 50% probability

−l with 50% probability

Then after N steps (time Nτ) in the random walk, the position (and displacement) x of the walker

is

x(N) =
NX

i=1

si

and the displacement squared is

x
2
(N) =

 
NX

i=1

si

!2

(1)

Let us now think about what the average distance the walker has moved is. It is immediately

obvious that with the equal probabilities to go left and right,

〈x(N)〉 = 0

that is, the average position will always be at the origin. But this does of course not mean that

the particle always is at zero. It means that the probability of finding the particle somewhere is

centered at x = 0, but naturally the probability distribution gets wider with increasing numbers of

steps N .

To get a handle on the broadening, let us consider the squared displacement, Eq. 1. We can rewrite

this as

x
2
(N) =

 
NX

i=1

si

!2

=

NX
i=1

si

NX
j=1

sj =

NX
i=1

s
2
i +

NX
i=1

NX
j=1
j 6=i

sisj

Then consider the pair sisj for a given pair i, j, j 6= i. This quantity will be

sisj =


+l2 with 50% probability

−l2 with 50% probability

so on average the sum over sisj will be zero! But on the other hand

s
2
i = l

2

independently of whether si is +l or −l!

Hence the average after N steps will be

〈x2
(N)〉 = l

2
N (2)

7.2.1.1. Coding the random walk

Before we proceed to look at deeper into the physical significance of this, let us see how we can do

a Monte Carlo simulation of this.

• There is nothing hard here: we simply generate N uniform random numbers between 0

and 1, and if the uniform number is below 0.5 we walk downwards, otherwise we walk

upwards.

Basics of Monte Carlo simulations, Kai Nordlund 2006 JJ J � I II × 7



• We then collect the result, and repeat for a large number ntimes times to get a

representative average.

Here is an implementation in Fortran90:

program randomwalk1d
implicit none

integer :: N,nwalks,iwalk,istep
integer :: x,seed
double precision :: xsum,xsqsum,dir

character :: buf*80
integer, external :: iargc
double precision, external :: uniformrand

_
if (iargc() < 3) then | PRELIMINARIES

print *,’Usage: randomwalk N nwalks seed’ |
STOP |

endif |
|

call getarg(1,buf); read (unit=buf,fmt=*) N |
call getarg(2,buf); read (unit=buf,fmt=*) nwalks |
call getarg(3,buf); read (unit=buf,fmt=*) seed |

|
print *,’Doing random walk to’,N,’steps’,nwalks,’ times’ _|

_
xsum=0.0; xsqsum=0.0; | RANDOM WALK
do iwalk=1,nwalks ! LOOP OVER WALKS |

|
! New walk |

x=0; |
do istep=1,N ! LOOP OVER STEPS IN WALK iwalk |

dir=uniformrand(seed) |
if (dir<0.5) then |

x=x-1 |
else |

x=x+1 |
endif |

enddo |
xsum=xsum+x |
xsqsum=xsqsum+x*x |

enddo _|

print *,’<x(N)>’,xsum/nwalks,’ <x2(N)>’,xsqsum/nwalks

end program randomwalk1d

When ran, this gave e.g. the following result:

beam.helsinki.fi randomwalk> nice randomwalk1d 100 100000 12313
Doing random walk to 100 steps 100000 times
<x(N)> -2.648000000000000E-002 <x2(N)> 100.260080000000

beam.helsinki.fi randomwalk> nice randomwalk1d 100 100000 7142
Doing random walk to 100 steps 100000 times
<x(N)> 2.734000000000000E-002 <x2(N)> 100.126840000000

and with a hundred times better statistics:

Doing random walk to 100 steps 10000000 times
<x(N)> -1.932800000000000E-003 <x2(N)> 99.9611160000000
Doing random walk to 100 steps 10000000 times
<x(N)> 1.772000000000000E-004 <x2(N)> 99.9962144000000

So we see that indeed 〈x(N)〉 is on average 0 and 〈x2(N)〉 ≈ N , as predicted by our analytical

derivation!

7.2.2. Continuum limit: the diffusion equation

[G+T 12.5; G+T 7A]

This basic random walk can be rewritten as a continuum diffusion equation by taking the limit in

which the lattice spacing l and the time step τ go to zero.

Let us begin by writing the random walk behaviour in terms of a so called master equation. Let

P (i, N) denote the probability that a walker is at site i after N steps. Since walkers have an equal

probability to walk left and right, it is clear that

P (i, N) = 1
2P (i + 1, N − 1) + 1

2P (i− 1, N − 1)

To get a continuum limit with familiar names for the variables, we can identify

t = Nτ and x = il

Now we can rewrite the previous equation as

P (x/l, t/τ) = 1
2P (x/l + 1, t/τ − 1) + 1

2P (x/l− 1, t/τ − 1)

but since the probability is independent of the length or time scales, we have

aP (x, t) = P (ax, t) or bP (x, t) = P (x, bt)

for any constants a, b. So we can multiply the equation with l and τ to obtain

P (x, t) = 1
2P (x + l, t− τ) + 1

2P (x− l, t− τ)

We rewrite this by subtracting P (x, t− τ) and dividing by τ

P (x, t)− P (x, t− τ)

τ
=

P (x + l, t− τ) + P (x− l, t− τ)− 2P (x, t− τ)

2τ
(3)

The left hand side already clearly resembles the definition of a derivative, if we take the limit

τ −→ 0.

To see what happens on the right hand side, we expand the P functions as Taylor series about x

and t with l and τ as the deviation. We only write out the terms needed:

P (x, t− τ) ≈ P (x, t)−
∂P (x, t)

∂t
τ

P (x± l, t− τ) ≈ P (x, t)±
∂P (x, t)

∂l
l + 1

2

∂2P (x, t)

∂x2
l
2 −

∂P (x, t)

∂t
τ

and we can rewrite eq. ?? as

P (x, t)− P (x, t− τ)

τ
≈

1

2τ
[P (x + l, t− τ) + P (x− l, t− τ)− 2P (x, t− τ)]

P (x, t)− P (x, t) + ∂P (x,t)
∂t τ

τ
≈

1

2τ

"
P (x, t) +

∂P (x, t)

∂l
l + 1

2

∂2P (x, t)

∂x2
l
2 −

∂P (x, t)

∂t
τ

+P (x, t)−
∂P (x, t)

∂l
l + 1

2

∂2P (x, t)

∂x2
l
2 −

∂P (x, t)

∂t
τ

−2P (x, t) + 2
∂P (x, t)

∂t
τ

–

and after we do all the obvious cancellations we get

∂P (x, t)

∂t
≈

l2

2τ

∂2P (x, t)

∂x2

In the limit τ → 0, l→ 0 but where the ratio l2/τ is finite, this becomes an exact relation. If we

further define D ≡ l2/2τ we get the equation

∂P (x, t)

∂t
= D

∂2P (x, t)

∂x2
(4)

But this is just the well-known macroscopic diffusion equation in 1 dimension, with D being exactly

the diffusion constant. So we have from a very simple microscopic model derived a well-known

macroscopic equation!

For future reference we note that one possible solution of this is a Gaussian of the form

P (x, t) =
N

√
4πDt

e
−x2/4Dt

(5)

(this can be easily shown by insertion of this form into the diffusion equation ??).

For those of you who do not know diffusion, what the equation does is describes the time dependence

of e.g. particle or heat diffusion in a system with a concentration or temperature gradient.

If we furthermore compare with the previous equation ??, 〈x2(N)〉 = l2N and use the identities

D ≡ l2/2τ and t = Nτ we can rewrite eq. ?? as

〈x2
(t)〉 = 2Dt (6)

This equation is called the Einstein relation after a certain physicist who first derived it back in

1905 [A. Einstein, Ann. Phys. 17 (1905) 549].

An alternative form of it is obtained by combining Eqs. ?? and ??:

l
2
N = 2Dt (7)

or using t = Nτ

D =
1

2

l2

τ
(8)

The 3-dimensional generalizations, which are straightforward to derive in a similar manner, are:

∂P (x, y, z, t)

∂t
= D∇2

P (x, y, z, t) (9)

and for the Einstein relation:

〈R2
(t)〉 = 6Dt (10)

where

〈R2
(t)〉 = 〈x2

(t) + y
2
(t) + z

2
(t)〉

and the factor of 2 has been replaced by a 6.

Furthermore, it even turns out that these results are not specific to being on a lattice, i.e. random

step lengths chosen in a suitable manner lead to the exact same result.

7.2.3. Proof that distribution becomes a Gaussian

We have earlier on this course mentioned several times the central limit theorem, which says that

many consequtive stochastic steps eventually lead to a Gaussian distribution.

• We will now show this explicitly for the 1D random walk on a lattice!

We consider walks with n = N steps of length l with a time between jumps τ , as above.

We want to know the probability that the particle at some time t = nτ is at a specific position x.

x

specific
x

One possible path which leads to x (steps
in y-direction only for illustration).

In this case n = 10, m = 2:
6 steps to the right, 4 to the left

0 +l-l

We need the following two variables:

n = t
τ : number of jumps the particle has made at time t.

m = x
l : number of steps to the right which corresponds to positions x.

It is obvious that the particle comes to x if it takes totally 1
2(n + m) jumps to the right and

1
2(n−m) jumps to the left:

1
2(n + m) to the right

1
2(n−m) to the left

n total steps.

What we want to know is pt(x) = pn(m) = the probablity to select among n total steps

specifically 1
2(n + m) steps to the right.

We shall now see stepwise how one can deduce the probability. First one must realize that 1
2(m+n)

must be a whole number. It is for instance impossible to take one step to the right if n = 2, and

vice versa if n = 3 it is impossible to take two steps to the right. Thus n and m must both be

either even or odd.

m is of course limited to the interval [−n, n].

The total number of possible random walks is of course 2n, so a prefactor of (1
2)

n can be used in

front of distinct choices of walks which lead to the same m.

To understand how the distribution forms up, we start by writing out all possible walks for increasing

n, for all the possible resulting m.

n m (1
2)(n + m) Possible walks Probability pn(m)

1 -1 0 ← (1
2)

1 1 1 → (1
2)

2 -2 0 ←← (1
2)

2

2 0 1 ←→;→← (1
2)

2 × 2× 1

2 2 2 →→ (1
2)

2

3 -3 0 ←←← (1
2)

3

3 -1 1 ←←→;←→←;→←← (1
2)

3 × 3
1

3 1 2 ←→→;→←→;→→← (1
2)

3 × 3×2
2×1

3 3 3 →→→ (1
2)

31

4 -4 0 ←←←← (1
2)

4

4 -2 1 ←←←→;←←→←;←→←←;→←←←; (1
2)

4 × 4
1

4 0 2 ←←→→;←→←→;→←←→; (1
2)

4 × 4×3
2×1

→←→←;→→←←;←→→←
4 2 3 ←→→→;→←→→;→→←→;→→→←; (1

2)
4 × 4×3×2

3×2×1

4 4 4 →→→→ (1
2)

4

Here the probabilities were intentionally written in a manner which makes one see how the total

probability forms up.

By looking at the results above one sees that in alol cases the probability can be written as

pn(m) = (
1

2
)
nn(n− 1)...[n− 1

2(n + m) + 1]

[12(n + m)]!
(11)

= (
1

2
)
n n!

[12(n + m)]![12(n−m)]!
(12)

This is the binomial distribution (and those who are well versed in statistics could have written out

the result directly).

By doing a lengthy but tedious derivation one can show that this leads to the same average and

mean square displacement as already obtained earlier, and finally to the Einstein relation

D =
1

2

l2

τ
(13)

But since we have already shown this exactly we do not repeat this here.

What we now want to do is get the continuum limit of the distribution.

To do this, we need the Stirlings approximation in the form

N ! '
√

2πN(
N

e
)
N

(14)

– can be derived using Γ functions, see e.g. Arfken-Weber.

This can be rewritten in logarithmic form

ln N ! ' N ln N −N +
1

2
ln(2πN) (15)

We now take the logarithm of both sides of

pn(m) =
1

2n

n!
1
2(n + m)!12(n−m)!

(16)

and get using Stirlings approximation

ln pn(m) = −n ln 2 + n ln n− n +
1

2
ln(2πn) (17)

−
1

2
(n + m) ln

„
1

2
(n + m)

«
+

1

2
(n + m)−

1

2
ln(π(n + m)) (18)

−
1

2
(n−m) ln

„
1

2
(n−m)

«
+

1

2
(n−m)−

1

2
ln(π(n−m)) (19)

The terms with only n and m cancel each other out,

We then divide terms of the type ln 1
2(n−m) into the form − ln 2 + ln(n−m), and obtain

= −n ln 2 + n ln n +
1

2
ln(2πn) +

1

2
(n + m) ln 2 (20)

−
1

2
(n+m) ln(n+m)−

1

2
ln(π(n+m))+

1

2
(n−m) ln 2−

1

2
(n−m) ln(n−m)−

1

2
ln(π(n−m))

(21)

Now the terms with ln 2 cancel each other out:

ln pn(m) = n ln n +
1

2
ln(2πn) (22)

−
1

2
(n+m) ln(n+m)−

1

2
ln(π(n+m))−

1

2
(n−m) ln(n−m)−

1

2
ln(π(n−m)) (23)

To get further we reorganize the stuff and use the Taylor approximation ln(1 + x) ≈ x− 1
2x

2:

ln(n + m) = ln n(1 + m/n) = ln n + ln(1 + m/n) (24)

' ln n + m/n−
1

2
(
m2

n2
) (25)

This is valid when |m| << n, i.e. far from the maximum possible value of n.

Similarly one can also do

ln(π(n + m)) = ln(πn(1 + m/n)) ' ln(πn) + m/n (26)

Terms of this type are guaranteed to be smaller than the other since they do not have an n or m

in front so for them we can skip the second-order term.

With these approximations we obtain

ln pn(m) ' n ln n+
1

2
ln(2πn)−

n

2
(1+m/n)

"
ln n + m/n−

1

2

m2

n2

#
−

1

2
[ln(πn) + m/n]

(27)

−
n

2
(1−m/n)

"
ln n−m/n−

1

2

m2

n2

#
−

1

2
[ln(πn)−m/n] (28)

The terms of the type 1/2m/n cancel, and we can sum up the terms which contain π:

1

2
ln(2πn)−

1

2
ln(πn)−

1

2
ln(πn) =

1

2
ln

2πn

πnπn
=

1

2
ln

2

πn
=

1

2
ln

4

2πn
= ln

2
√

2πn
(29)

We use this result and divide the rest into parts which depend on n and m:

= n ln n−
n

2

"
ln n +

m

n
−

1

2

m2

n2

#
−

m

2

"
ln n + m/n−

1

2

m2

n2

#
(30)

−
n

2

"
ln n−m/n−

1

2

m2

n2

#
+

m

2

"
ln n−m/n−

1

2

m2

n2

#
+ ln

2
√

2πn
(31)

The quadratic terms m2/n2 with m in front of the square brackets cancel and obtain

ln pn(m) = n ln n−
n

2

"
ln n +

m

n
−

1

2

m2

n2

#
−

m

2
[ln n + m/n] (32)

−
n

2

"
ln n−m/n−

1

2

m2

n2

#
+

m

2
[ln n−m/n] + ln

2
√

2πn
(33)

= +
m2

2n
−

m2

n
+ ln

2
√

2πn
= −

m2

2n
+ ln

2
√

2πn
(34)

Hence

pn(m) '
2

√
2πn

e
−m2/2n

(35)

This is obviously a Gaussian form, and thus this completes the proof that the random walk leads to

a Gaussian!

But it is still worth rewriting this into a more familiar form.

Let us recall the definitions of m and n:

m =
x

l
(36)

n =
t

τ
(37)

Using these we obtain

p(t)(x) =
2q
2π t

τ

e
−x2

l2
τ
2t (38)

which can be written as

p(t)(x) =
2q
2πt
τ

e
−x2

t ( τ
2l2

)
(39)

To reiterate, this is the probability to have a particle at x, a dimensionless quantitity. To obtain the

particle density particles/length n we divide this with our length scale l and get

n(t)(x) =
2

l
q

2πt
τ

e
−x2

t ( τ
2l2

)
(40)

If we now compare this with the continuum theory solution for the diffusion equation in one

dimension, eq. ??

n(x, t) =
N

√
4πDt

e
−x2/4Dt

(41)

we see that these are identical if we use the Einstein-relation as earlier, Eq. ??

D =
1

2

l2

τ
(42)

Hence now we have for the 1D random walk shown that in the limit of large number of steps it

leads to identical behaviour as macroscopic diffusion in all respects:

• Same average = 0

• Same mean square displacement

• Same distribution

This provides very strong proof that the macroscopic phenomenon of diffusion can be explained

with microscopic movement of matter!

Here is an illustration of how the simulated solution approaches the exact Gaussian distribution for

increasing n.

Note that already for n = 10 we are quite close to the Gaussian!

7.2.4. Microscopic and macroscopic diffusion

What is the relation between using MC simulations to deal with a random walk and the macroscopic

equation? This depends on the outlook.

• If we are interested in macroscopic diffusion, we can use MC simulations of random walk

essentially as an advanced numerical means to solve the diffusion equation ??.

• Of course ordinary numerical solution of partial differential equations is in many cases

likely to be perfectly adequate for dealing with diffusion, and much more efficient than

random walk simulations.

– But the more complicated the diffusion problem gets, the harder it becomes to set

up and solve the diffusion equation, and at some point it may actually become both

easier and more efficient to use a microscopic random walk model.

– By utilizing the kinetic Monte Carlo approach (dealt with in the next section of

this course) to random walks the microscopic simulations actually can become quite

efficient.

On the other hand, if we are primarily interested in understanding the microscopic processes

themselves, we can do random-walk like simulations on the microscopic level to see what is really

going on.

• The relation with the diffusion equation can then just be used check that the microscopic

mechanisms will lead to the expected macroscopic limit if we expand our number of

walkers towards infinity and look at larger length scales.

For instance, if we look at atom migration, the l can naturally be the known interatomic distance,

and τ the average jump frequency of the atoms.

• In this case taking the limit l → 0 and τ → 0 has no physical meaning since the real l

and τ are finite.

• Instead we can think of obtaining the macroscopic, continuum diffusion equation by

looking at time scales t >> τ and length scales L >> l.

The Einstein relation is extremely useful for this kind of problems in that it allows determining the

macroscopic diffusion constant in a simple way from microscopic data of ∆x and τ .

• In the pure random walk as the ones here, there is actually not much information to gain

as one has to know τ in advance.

• But if for instance MD simulations are used to predict the defect motion, it is possible to

predict the diffusion coefficient.

It is also interesting to note that present-day physics very often deals with truly atomistic length

scales — in commercial Si microprocessors the smallest parts are already today only a few ten atom

layers thick.

• In this kind of system, the assumption that the system length scale L >> l may no

longer be true, and the continuum approximation starts to become increasingly inaccurate.

• In such cases a random walk look on migration may be the most appropriate way to deal

with diffusion, and MC simulations of random walk the best tool to work with theoretically.

• This has now lead to the semiconductor industry being directly involved in atom-level

diffusion simulations.

7.3. Modified random walks

In addition to the completely isotropic random walk described above, there are numerous variation

where the walk directions are non-isotropic or constrained somehow.

7.3.1. Scaling law

In considering other types of random walks, one often uses the concept of a scaling law to describe

the basic properties of the generator. For the basic random walk, we considered the dispersion

〈∆x
2
(N)〉 = 〈x2

(N)〉 − 〈x(N)〉2

and found that

〈∆x
2
(N)〉 = 〈x2

(N)〉 ∝ N
1

where N is the number of steps.

In other, more complicated walks it need not be true that 〈x(N)〉 = 0, nor does the exponent on

N need to be 1. Hence one can write for a general random walk that

〈∆x
2
(N)〉 ∝ N

2ν

and use the exponent ν to characterize the walk. A similar scaling law dependence can also be valid

for other quantities f of interest. It is customary to have 2ν in the exponent, because
√

f then

gives ν directly.

To determine the exponent in practice, it is often useful to make statistics of 〈∆x2(N)〉 as a

function of N while N is growing in the simulation. The one can output f(N) and use a fitting

routine to determine the exponent.

(Hint for novices in data analysis: a useful trick to do an extremely quick test of whether

something follows a scaling law: if f(N) is plotted in a log-log graph, it will appear linear if it

follows a power law. The slope gives the exponent. For xgraph the options -lnx -lny give a

log-log plot).

7.3.2. Examples

[G+T 12.2 –]

In here we will first list a few common types of modified random walks, very briefly describing what

they are. In the next two subsections we will look at two examples in greater detail, also describing

what real-life cases they correspond to.

7.3.2.1. Nonequal probabilities

It is of course possible to forgo the requirement that a step in all directions is taken with equal

probability. Say for instance that in the 1-dimensional example the probability to move right is p

and to the left is q = 1− p. This will introduce a drift in the system, and we will have

〈x(N)〉 = (p− q)lN

instead of 0. This is called a biased random walk, and corresponds to the macroscopic diffusion

equation with an extra term which can be interpreted as a velocity:

∂P (x, t)

∂t
= D

∂2P (x, t)

∂x2
− v

∂P (x, t)

∂x
(43)

The physical interpretation can e.g. be a raindrop falling in a swirling breeze.

7.3.2.2. Persistent random walks

In the simple random walk, the probability that a step is taken into a certain direction is independent

of the previous steps, and in fact constant all the time. In case the jump probability depends on the

previous transition, one can talk about a persistent random walk.

A simple example is again given in a 1D system: if a jump has just been made at step N − 1, let us

say that the probability that the next jump occurs in the same direction is α, and the probability

it occurs in the opposite direction is 1− α.

Basics of Monte Carlo simulations, Kai Nordlund 2006 JJ J � I II × 34



– In case α = 0.5 we have just the ordinary random walk.

• But if α 6= 0.5 the two jumps will be correlated.

• If α > 0.5 there is a simple physical interpretation of the correlation: once an object

starts moving, it is more likely it will keep moving in the same direction on successive

steps.

– This has e.g. been observed in the study of molecules in a chromatographic column.

These molecules can either be in an immobile, trapped phase with zero velocity, or in

a mobile phase with velocity v.

– For steps of unit length the position will then change at each step by v or 0.

– But once the molecule is in a mobile state, it tends to stay there, so one could describe

the motion using an α > 0.5 to determine how it behaves.

7.3.2.3. Restricted random walks

Another important modification of the walk is to consider a system with certain points or regions

which somehow restrict the motion of the atom. Such a restriction could e.g. be a line which

reflects particles, a region which absorbs them or that two particles can not occupy the same site.

The simplest possible restriction is probably a trap in a lattice. It is simply a site which absorbs

every random walker it encounters.

Traps are important in condensed matter physics. For instance, we can consider a simple model for

light energy transport in a solid.

• A lattice is considered to have “hosts” and traps. When an incoming photon reaches a

host, the light energy will be transferred into an electron excitation, which can travel in

the lattice somewhat like a particle, an “exciton”.

• When the exciton reaches a trap site it is absorbed there, and a chemical reaction can

occur there due to the added energy from the exciton.

For defects in solids, a surface often acts as a “trap line” which absorbs all defects which reach it.

In subsection ?? below I describe in detail how to model a system where particles can not occupy

the same site.

7.3.2.4. Continuum walks

There is no reason why a random walk should only be on a lattice with a constant step size. If you

e.g. think about atom motion in a gas, it is quite clear that the distance between collisions is not

very likely to be constant, and the atoms can of course move in any direction.

There are continuum walks, where the steps can be in any direction. The step length may be fixed,

or of variable length with some distribution.

As a simple example, if you do a 2D random walk in a random direction with steps of unit length,

it turns out that for large N the distribution of displacements will become a Gaussian.

Finally, it is of course also possible to have a walk on a lattice where the displacement may be

longer than to the nearest neighbour sites. Certain interstitial atom diffusion mechanisms can have

such a character 1.

1For those of you who know about diffusion in metals, I am thinking of crowdion motion in BCC metals

7.4. Lattice gas

Consider a finite lattice with some density ρ of Np particles. The particles can move on the lattice

by jumps to the nearest sites, but two particles can not occupy the same site. This is a simple

example of a restricted random walk (see above). The physical interpretation is e.g. vacancies

moving in a lattice.

To simulate this kind of system, we need a bit more of an advanced approach than before.

• First of all, we need to simulate the motion of all the particles at the same time, not

taking the average over many independent single-particle motions as was done before.

• To be able to meet the criterion that two particles should not occupy the same site, we

can do two things.

– One is to make an array which contains all possible lattice sites. The other is to, at

each move, find the distance to all other particles and check that no one occupies the

site to be moved to.

– In case the lattice is small enough to fit the available computer memory, using the

former solution is much easier and faster.

In case a particle jumps on average after every time ∆t, then if we have Np particles we should

make a move of one of the particles at time intervals of

∆tp = ∆t/Np

But it is possible that a particle can not move at all, if all neighbouring sites are occupied. Should

the time in this case be advanced by ∆tp or not? This depends on the physical situation we are

involved with. In this case, we will choose to advance the time.

Another choice to make is whether we should always just try a jump in one direction, or for a given

particle choice choose one of the available jumps if any is available. This time, we choose the latter

one, because it is harder to code.

Finally, we have to choose how to treat the borders. Now we will choose to use periodic
boundaries. This simply means that if a particle exits the boundary on one side, it comes out on

the other side. This mimics a section of an infinite system.

There is one non-trivial practical thing to consider when coding this. That is that we can not now

simply evaluate

∆x = x(t = 0)− x(t)

because the particles may move several times over the periodic boundaries around the cell. In this

case the above equation could give a much too small answer. What we have to do is introduce

another array dx which sums up the displacement of the particle, before consideration of the

periodic boundaries.

(Note that if you read Gould-Tobochnik about this, their code does not take account of this!)

So to summarize, here is a sketch of an algorithm on how to simulate this kind of a walk.

1◦ Choose number of particles Np, number of steps Nsteps, side length L. Set ∆t and lattice

size a.
2◦ Set all positions in the L× L grid to be empty

3 a◦ Generate Np particle coordinates randomly on the grid, checking that no two particles end

up on the same points.

3 b◦ Mark the points with the particles in the L× L grid as filled.

4◦ Loop over MC steps of time ∆t

5◦ Loop from 1 to Np

6◦ Pick one particle i at random

7◦ Find which positions it can jump to. If none, return to step 6◦

8◦ Let the particle jump to one of the allowed directions j by a displacement

xi = xi + δxj, yi = yi + δyj, enforce periodic boundaries on x and y

9◦ Set dxi = dxi + δx, dyi = dyi + δx (where periodic boundaries do not play a

role!)

10◦ End loop from 1 to Np

11◦ Update time t = t + ∆t

12◦ End loop over MC steps

13◦ Output 〈∆R
2〉 = 〈dx

2
i + dy

2
i 〉 and calculate diffusion coefficient.

The crucial difference here to the previous random walk algorithms is that the outer loop goes

over MC steps, the inner one over particles. When the walkers are independent of each other

(“non-interacting”) we can deal with one walker at a time, saving memory since storage of all

particles is not needed.

How to do this in practice is illustrated in the Fortran90 code below. It gives out the diffusion

constant D as the final answer, having reasonable values of 1 ns for the average jump time and 2

Å for the jump distance.

Note that this program fails for large numbers of particles using the Park-Miller “minimal
standard” random number generator. In that case, the diffusion coefficient will keep growing

instead of stabilizing at large times. The code below used the Mersenne twister, the Fortran version

available from the course home page. (Yes, I did find this out the hard way, spending almost an

entire Sunday debugging my code before I realized the random number generator was the culprit!)

! To compile use e.g.
! for Linux/Absoft Fortran:
! f90 -O -o randomwalk_latticegas randomwalk_latticegas.f90 -lU77 -lfio
! For Alphas with Compaq Fortran:
! f90 -O -o randomwalk_latticegas randomwalk_latticegas.f90
!

program randomwalk_latticegas
implicit none

logical, allocatable :: lattice(:,:)
integer, allocatable :: x(:),y(:)
double precision, allocatable :: dx(:),dy(:)

integer :: Nsteps,Np,L,narg
integer :: istep,isubstep
integer :: dir,i,j,seed,nfail,njumps

integer, parameter :: MAXINT=1000000000

! Variables for counting allowed directions
integer :: free(4),nfree
integer :: dxtrial(4),dytrial(4)
integer :: xnew(4),ynew(4)

double precision :: dxsum,dysum,dxsqsum,dysqsum
double precision :: t,deltat,drsqave,D,a,help

character :: buf*80
integer, external :: iargc
double precision, external :: uniformrand,grnd

! Set average time between jumps and jump length Units is s and cm
! although actually this is not needed for the simulation
deltat=1d-9; ! 1 ns
a=2e-8; ! 2 Å

if (iargc() <2) then
print *,’Usage: randomwalk_latticegas Nsteps Np L seed’

STOP
endif

call getarg(1,buf); read (unit=buf,fmt=*) Nsteps
call getarg(2,buf); read (unit=buf,fmt=*) Np
call getarg(3,buf); read (unit=buf,fmt=*) L
call getarg(4,buf); read (unit=buf,fmt=*) seed

call sgrnd(seed)

print *,’Doing lattice gas walk to’,Nsteps,’MC steps, initial seed’,seed
print *,’using’,Np,’ particles on a’,L,’^2 square lattice’

if (Np >= L*L) then
print *,’Are you crazy? Number of particles >= number of sites’
STOP ’Too small lattice’

endif

print *,’Allocating arrays’
allocate(lattice(0:L-1,0:L-1))
allocate(x(Np),y(Np))
allocate(dx(Np),dy(Np))

! Mark all positions as empty
do i=0,L-1

do j=0,L-1
lattice(i,j) = .false.

enddo
enddo

! enumeration of directions: 1 left 2 right 3 up 4 down

dxtrial(1)=+1; dytrial(1)= 0;
dxtrial(2)=-1; dytrial(2)= 0;
dxtrial(3)= 0; dytrial(3)=+1;
dxtrial(4)= 0; dytrial(4)=-1;

nfail=0; njumps=0;
! Generate particles on lattice
do i=1,Np

do ! Loop until empty position found
! To be on safe side, check that upper limit not reached
x(i)=int(grnd()*L); if (x(i)>=L) x(i)=L-1;
y(i)=int(grnd()*L); if (y(i)>=L) y(i)=L-1;
if (lattice(x(i),y(i))) then

! Position already filled, loop to find new trial
cycle

else
lattice(x(i),y(i))=.true.
! Success, go to next particle
exit

endif
enddo
dx(i)=0.0d0; dy(i)=0.0d0;

enddo

t=0.0;
do istep=0,Nsteps-1 ! Loop over MC steps

do isubstep=1,Np ! Do all particles on average once every MC step

! Pick one particle at random
i=int(grnd()*Np)+1; if (i>Np) i=Np;

! Find possible directions, store it in free()
nfree=0
do j=1,4

xnew(j)=x(i)+dxtrial(j);
if (xnew(j) >= L) xnew(j)=0; if (xnew(j)<0) xnew(j)=L-1;
ynew(j)=y(i)+dytrial(j);
if (ynew(j) >= L) ynew(j)=0; if (ynew(j)<0) ynew(j)=L-1;
if (.not. lattice(xnew(j),ynew(j))) then

! Success: position free
nfree=nfree+1
free(nfree)=j

endif
enddo
! If no possible directions, get new particle
if (nfree == 0) then

nfail=nfail+1
cycle

endif
njumps=njumps+1

! Pick one of the possible directions randomly
! Note that the dir>nfree check here really is needed!
dir=int(grnd()*nfree)+1; if (dir>nfree) dir=nfree
j=free(dir)

! Now x(i),y(j) is old position and xnew(j),ynew(j) new
! Double check that new site really is free
if (lattice(xnew(j),ynew(j))) then

print *,’HORROR ERROR: THIS SHOULD BE IMPOSSIBLE’
print *,i,j,dir,nfree
print *,free

print *,x(i),y(i),xnew(j),ynew(j)
STOP ’HORROR new site bug’

endif
!Empty old position and fill new
lattice(x(i),y(i))=.false.
lattice(xnew(j),ynew(j))=.true.

x(i)=xnew(j); y(i)=ynew(j);
dx(i)=dx(i)+dxtrial(j); dy(i)=dy(i)+dytrial(j);

enddo

if (mod(istep*Np,1000000) == 0) then
! Calculate and print intermediate results every now and then
! Get total displacement from dx,dy
dxsum=0.0d0; dysum=0.0d0;
dxsqsum=0.0d0; dysqsum=0.0d0;
do i=1,Np

dxsum=dxsum+dx(i); dysum=dysum+dy(i);
dxsqsum=dxsqsum+dx(i)*dx(i); dysqsum=dysqsum+dy(i)*dy(i);

enddo
drsqave=(dxsqsum+dysqsum)/(1.0*Np)

if (t>0.0) then
! Get diffusion coefficient by proper scaling
D=drsqave*a*a/(4*t)
print *,’At’,t,’ drsqave’,drsqave*a*a,’ D’,D,’ cm^2/s’

endif

endif

t=t+deltat

enddo

! Get total displacement from dx,dy
dxsum=0.0d0; dysum=0.0d0;
dxsqsum=0.0d0; dysqsum=0.0d0;
do i=1,Np

dxsum=dxsum+dx(i); dysum=dysum+dy(i);
dxsqsum=dxsqsum+dx(i)*dx(i); dysqsum=dysqsum+dy(i)*dy(i);

enddo
print *,’dxsum’,dxsum,’ dysum’,dysum
print *,’dxsqsum’,dxsqsum,’ dysqsum’,dysqsum
drsqave=(dxsqsum+dysqsum)/(1.0*Np)
print *,’drsqave’,drsqave
print *,’Number of failed jumps’,nfail,’ number of successes’,njumps
! Get diffusion coefficient by proper scaling
D=drsqave*a*a/(4*t)
print *,’At’,t,’ drsqave’,drsqave*a*a,’ D’,D,’ cm^2/s’

end program randomwalk_latticegas

((After this the Mersenne twister source code should follow.
In that code, you have to change the comment character from
‘‘*’’ to the Fortran90 ‘‘!’’.))

This will be animated during the lecture.

[[Lecturers own reminder on animation, reader can ignore:

cd opetus/mc/tests/randomwalk

f90 randomwalk_latticegas_output.f90 -lU77
a.out 100 2 20 12278 | grep "^ P" | dpc msleep 100 x -1 21 y -1 21 m 1 d 21 sd 440 440 erase 2 3 4 5 _

and then increase second argument.]]

And here is a series of results:

Np L Np/L^2 D (cm^2/s) nfail njumps
--- ---- ------ ------------- -------- --------
10 100 0.001 9.769973881166823E-008 0 10000000
10 100 0.001 1.127346430730184E-007 0 10000000
100 100 0.01 1.028685543050629E-007 0 10000000
100 100 0.01 9.469519884885580E-008 0 10000000
10000 1000 0.01 9.899003879678247E-008 0 10000000
1000 100 0.1 9.111043889255736E-008 292 9999708
1000 100 0.1 9.427090885414200E-008 279 9999721
100000 1000 0.1 9.403952985695557E-008 3127 99996873
3000 100 0.3 8.284148565973272E-008 109626 29890374
3000 100 0.3 7.915751903784448E-008 110196 29889804
6000 100 0.6 5.895798261670045E-008 1152902 10847098
6000 100 0.6 5.913229928124830E-008 1154808 10845192
9000 100 0.9 1.771291645136659E-008 11574471 6425529
9000 100 0.9 1.786338311620434E-008 11571431 6428569
900000 100 0.9 1.824779088931029E-008 57886778 32113222
9900 100 0.99 1.831247452488705E-009 19013835 786165
9900 100 0.99 1.860272704661156E-009 19015892 784108

What does this mean? At small concentrations, the system behaves essentially as an unconstrained

random walk. For that one, we know that 〈∆R2〉 should be equal to a2N , where N is the number

of steps, and a is the jump distance, and the result for the diffusion coefficient should be

D =
〈∆R2〉

4t
=

(2 Å)2N

4N∆t
=

(2 Å)2

4× 1 ns
= 10

−7 cm2

s

which is exactly what we get, within the uncertainty. But for larger concentrations, the number

of failed jumps starts to grow, which reduces the diffusion coefficient. It is interesting to note,

however, that the diffusion coefficient and number of failed jumps do not follow an obvious linear

dependence, so doing the MC simulation really is worthwhile.

Here is still a plot of D(ρ), where ρ is the particle concentration.

7.5. Length of polymers in a good solvent

[Gould-Tobochnik 12.3]

As a final example of where random walks can be useful, we consider the length of polymers lying

in a good solvent. A polymer is a molecule which consists of a long chain of basic building blocks.

For instance, polyethylene can be represented as

· · ·–CH2–CH2–CH2–CH2–· · ·

When a polymer is placed in a good solution, where it basically can float around freely, it will after

some time adopt a twisted form which is neither straight nor completely curled up. Imagine a single

noodle (spaghetti strain) in warm water.

Experiments on polymers have shown that when the length of the “noodle” is measured, the

average mean square end-to-end length has been found to have the following dependence:

〈∆R
2
(N)〉 ∝ N

2ν
with ν ≈ 0.592

for a wide range of polymers, regardless of their other properties. Here N is the number of

monomers. The fact that such very general behaviour is observed indicates that there is some very

simple general reason for the behaviour.

This lead Rosenbluth and Rosenbluth [Ref. J. Chem. Phys 23 (1955) 356 according to G+T] to think that a very

simple random walk approximation could explain this behaviour. The main idea here is as follows.

Let us think of making a random walk whose path would give the shape of the particular polymer.

The figure below illustrates a polymer and a random walk which would approximate the shape of

this polymer.

A simple random walk clearly can not explain the behaviour, since it gives an exponent of ν = 0.5.

• But it is also clear that the ordinary random walk does not describe the physical situation

sensibly either, because it can return to the same positions many times.

• But in a polymer the monomers certainly can not occupy the same space – atoms can

just not be squeezed on top of each other without extreme kinetic energies or pressures.

• So let us modify the basic random walk as follows: we add the simple requirement that

the walker should never return to a position it has already occupied. This kind of random

walk is called the self-avoiding walk (SAW).

If we know what the minimum-energy angles between the monomers are, we could take these angles

to construct the walk.

• But since the same scaling behaviour is observed for many kinds of polymers, this suggests

that the result will be fairly independent of the angles.

• So we can start with using a square lattice (2D) or simple cubic lattice (3D) with all

angles of 90◦ to simulate the walk. And it turns out that this gives almost exactly the

right answer!

This is a really beautiful example of how a very simple physical model can lead to understanding of

nature!.

7.5.1. Coding self-avoiding random walks

Coding the SAW efficiently is not quite trivial. It could seem easy. Just do random walks one at a

time as usual for non-interacting walkers, store all the positions visited to an 2D or 3D array, and if

a new step enters a previously visited site, disregard it.

But herein lies the problem. We can not just disregard the last step, and look for a new direction.

This would skew the statistics; we have to disregard the whole walk. If we want to make a walk of

any sensible length (say, N > 30), the probability of the walker entering a previously visited site

becomes enormously large. Hence we end up disregarding almost all walks.

In my tests, in 3D walks for N = 20 only roughly 2 % of the walks were successful, for N = 30

only 0.1%, and for N = 40 only 0.04%. So the efficiency is terrible even for small N , and quickly

becomes truly horrendous for increasing N .

Hence it is very important to have some way to make the simulation of the SAW’s more efficient.

We will here present the simplest one, the optimization scheme of Rosenbluth and Rosenbluth.

The idea is that once a walk attempts to do a step N that is impossible, we do not disregard the

whole walk, but only this step. We then pick some other, possible step. A weight factor W (N) is

used to ensure that the statistics remains correct.

The way the weighting is done can be written as follows. Consider a situation where we have just

done step N − 1, and look at step N . Three outcomes are possible:

1◦ No step is possible. Set W (N) = 0 and restart

2◦ All steps other than the step right backward are possible, set W(N)=W(N-1)

3◦ Only m steps are possible with 1 ≤ m < 3 (2D) or 1 ≤ m < 5 (3D). In this case

we choose randomly one of the possible steps, and set W (N) = m
3 W (N − 1) (2D) or

W (N) = m
5 W (N − 1)(3D).

The correct value of the final average 〈∆R2(N)〉 is obtained by weighting R2
i (N) for each step

N in a given walk i with the weight Wi(N) obtained in the same walk. To get the average, we

have to divide by the sum of Wi(N) instead of N , i.e.

〈∆R
2
(N)〉 =

NwalksX
i=1

Wi(N)R
2
i (N)

NwalksX
i=1

Wi(N)

W (0) is initialized to 1.0.

This is tremendously much more efficient than the direct approach because it is actually quite rare

to end up in a situation where no step at all is possible. Hence the number of failures is vastly

smaller than before: in my tests I obtained only 8 failures for 3D walks of length 30 simulating

10000 walks.

Note that in handling step 1 there are actually two choices. One is disregarding the whole walk,

the other just disregarding the steps after the N we have managed to reach.

If the goal is to deal with walks (polymers) always reaching the full length, one should do the

former. But which one to do may actually depend on how the experiments are exactly carried out.

But as long as there are very few failed walks, this does not in any case make much of a difference.

There are other, more advanced ways of speeding up the SAW (see e.g. Gould-Tobochnik). But

already the Rosenbluth-Rosenbluth scheme is easily good enough to test the basic result of whether

ν = 0.592.

